

Tutorial

Berechnung der Rangreihenfolge mit Excel

Hermann Baum

Januar 2026

INHALT

Einleitung.....	2
Teil 1:.....	3
Kriterien Punktzahl, Tordifferenz und erzielte Tore.....	3
Teil 2.....	5
Direkte Vergleiche	5
2.1 Umsetzung des direkten Vergleichs in Formeln	6
2.2 Zerlegung der Funktion ISTZAHL in ihre Einzelteile	7
2.3 Abschluss des ersten Direktvergleichs	8
2.4 Zweiter Direktvergleich	9
Teil 3.....	10
Weitere Bemerkungen zur Excel-Datei	10

Einleitung

Bei der Bestimmung der Rangreihenfolge innerhalb einer Gruppe sind in der Regel mehrere Kriterien mit unterschiedlichen Prioritäten zu berücksichtigen. Im einfachsten Fall sind dies die Anzahl der Punkte, die Tordifferenz und die Anzahl der erzielten Tore.

Meistens kommen noch die direkten Vergleiche hinzu, die bei der Fußball EM der UEFA und erstmals bei der Fußball-WM 2026 sogar mehrfach angewendet werden. Da wird die Umsetzung in Excel schnell zu einem undurchdringlichen Formelgewirr.

In diesem Tutorial soll nun eine elegante, vergleichsweise schlanke und sehr übersichtliche Art vorgestellt werden, die Rangreihenfolge innerhalb einer Gruppe zu berechnen.

Gerade die direkten Vergleiche erfordern viel Aufwand – insbesondere wenn es sich um mehr als vier Teams pro Gruppe handelt. Bei 9 Teams pro Gruppe sind z.B. bei meinen bisherigen Turnierplänen auf dem dafür zuständigen Tabellenblatt ‚Calc‘ mehr als 1000 Matrixformeln (pro Gruppe!) im Einsatz. In der hier vorgestellten Methode muss man nur eine einzige Matrixformel verstehen, die bei 9 Teams in der Gruppe 27-mal zum Einsatz kommt bzw. 54-mal bei wiederholtem Direktvergleich.

Das Tutorial ist in zwei Teile aufgeteilt. Im ersten Teil wird die Berücksichtigung der Kriterien Punkte, Tordifferenz und erzielte Tore besprochen. Die Arbeit dafür erledigen drei einfache Formeln, eine mit der RANG-Funktion, eine mit der ZÄHLENWENNS-Funktion und eine dritte, die einfach die Summe zweier Zellen berechnet.

Der zweite Teil beschäftigt sich mit den direkten Vergleichen. Er beleuchtet die Kriterienreihenfolge der UEFA bzw. der Fußball-WM 2026: Punktzahl, Direktvergleich 1 (Punktzahl, Tordifferenz, erzielte Tore), Direktvergleich 2 (Punktzahl, Tordifferenz, erzielte Tore), Tordifferenz aus allen Spielen, erzielte Tore aus allen Spielen, Fair-Play.

Bei den direkten Vergleichen kommt eine etwas komplexere Matrixformel zum Einsatz, die aber in ihre Einzelteile zerlegt und ausführlich erklärt wird.

Eine grundlegende Einführung in Matrixformeln gibt es → [hier](#).

Zum Tutorial gibt es eine zugehörige Exceldatei: → [Excel-Datei herunterladen](#)

Teil 1:

Kriterien Punktzahl, Tordifferenz und erzielte Tore

	C	D	H	I	J	K	L	M	N	O	P	Q	R	S	T	U	V	W	X
1	de 2:	FALSCH																	
2													RANK		GD	RANK		GF	RANK
3			GF	GA	GD	Pts	Pld	W	D	L	Fair play		cr.1		add	cr.2		add	cr.3
4	1	1. FC Riedwald	8	7	1	6	5	2	0	3	0		2		0	2		0	2
5	2	FC Barcelona	10	11	-1	6	5	2	0	3	0		2		3	5		0	5
6	3	Eintracht Frankfurt	10	10	0	6	4	2	0	2	0		2		1	3		0	3
7	4	Maibach 1822 eV	3	2	1	4	2	1	1	0	0		8		0	8		0	8
8	5	VFB Essenheim	11	8	3	9	5	3	0	2	0		1		0	1		0	1
9	6	Inter Mailand	3	7	-4	6	4	2	0	2	0		2		5	7		0	7
10	7	SSV Wildbach	2	1	1	4	2	1	1	0	0		8		0	8		1	9
11	8	SV Obersuhl	10	11	-1	6	5	2	0	3	0		2		3	5		0	5
12	9	SSC Bosserode	10	10	0	6	4	2	0	2	0		2		1	3		0	3

Abb. 1.1

Die **Grundidee** in beiden Teilen ist, dass nach jedem Kriterium eine Rangfolge erzeugt wird, die die Reihenfolge nach Anwendung dieses Kriteriums zeigt.

So sieht man in der Spalte R die Rangfolge nach Anwendung des ersten Kriteriums, der Punktzahl. Die Zahlen in Spalte R nennen den Rang des jeweiligen Teams. Das erste Team (das Team in der obersten Zeile: 1. FC Riedwald) steht nach Vergleich der Punktzahlen auf Rang 2, das zweite und dritte Team ebenfalls auf Rang 2, das vierte Team auf Rang 8 usw.

Team 5 (VFB Essenheim) ist mit Rang 1 Tabellenführer. Es hat als einziges Team 9 Punkte erreicht. Sechs Teams belegen gemeinsam den Rang 2. Sie haben jeweils 6 Punkte erreicht. Diese Rangzahlen in Spalte R werden durch folgende einfache Formel erzeugt:

=RANG(\$K4;\$K\$4:\$K\$12)

=RANG(\$K5;\$K\$4:\$K\$12)

...

=RANG(\$K12;\$K\$4:\$K\$12)

Um nun das zweite Kriterium, die Tordifferenz, ins Spiel zu bringen, werden in der Spalte T die folgenden Formeln eingetragen:

=ZÄHLENWENNS(\$J\$4:\$J\$12;">"&\$J4;\$R\$4:\$R\$12;"="&\$R4)

=ZÄHLENWENNS(\$J\$4:\$J\$12;">"&\$J5;\$R\$4:\$R\$12;"="&\$R5)

=ZÄHLENWENNS(\$J\$4:\$J\$12;">"&\$J6;\$R\$4:\$R\$12;"="&\$R6)

...

=ZÄHLENWENNS(\$J\$4:\$J\$12;">"&\$J12;\$R\$4:\$R\$12;"="&\$R12)

Die ZÄHLENWENNS-Funktion prüft zwei Bedingungen.

Erste Bedingung: \$J\$4:\$J\$12;">"&\$J4

Es wird in jeder Zeile geprüft, ob die Tordifferenz des Teams in dieser Zeile besser ist als diejenige des betreffenden Teams.

Zweite Bedingung: \$R\$4:\$R\$12;"="&\$R4

Es wird in jeder Zeile geprüft, ob die Rangzahl in Spalte R dieselbe ist wie diejenige des betreffenden Teams.

Das heißt, die Formel zählt für das betreffende Team, wie viele Teams mit derselben Rangzahl eine bessere Tordifferenz haben. Zum Beispiel steht in Zelle T9 eine „5“. Das bedeutet für das betreffende Team (Inter Mailand), dass fünf Teams gezählt wurden, die auf demselben Rang stehen und eine bessere Tordifferenz haben, nämlich die Teams 1, 2, 3, 8 und 9.

Das Praktische an dieser Zahl ist, dass sie genau angibt, um wie viele Ränge das betreffende Team (Inter Mailand) nach unten rutscht durch die schlechtere Tordifferenz. Zur bisherigen Rangzahl 2 braucht man also nur die 5 aus der Zelle T9 addieren. Und das erledigt die Formel in Zelle U9.

Sie lautet: =R9+T9

So entsteht entsprechend der oben genannten Grundidee eine neue Liste von Rangzahlen, die die Reihenfolge nach Anwendung des Kriteriums ‚Tordifferenz‘ zeigt.

Genauso funktionieren die Formeln in den Spalten W und X, die die Rangfolge nach Anwendung des dritten Kriteriums (Anzahl der erzielten Tore) erzeugen. Die Formel in Spalte W bezieht sich auf die Spalte H, wo die erzielten Tore zu finden sind.

Hier fällt Team 7 (SSV Wildbach) auf Rang 9 zurück, weil Team 4 (Maibach 1822 eV) ein Tor mehr erzielt hat.

In Spalte X haben wir somit die Rangreihenfolge nach Anwendung der ersten drei Kriterien.

Team 3 (Eintracht Frankfurt) und Team 9 (SSC Bosserode) stehen immer noch auf demselben Rang, nämlich Rang 3, ebenso Team 2 (FC Barcelona) und Team 8 (SV Obersuhl), die beide auf Rang 5 stehen.

Hier würden direkte Vergleiche folgen, die aber in diesem speziellen Fall keine Entscheidung bringen, weil diese Teams noch nicht gegeneinander gespielt haben.

Da diese drei Teams somit im ersten Direktvergleich nicht unterscheidbar sind, werden sie einem zweiten direkten Vergleich unterzogen, in dem nur die Spiele dieser drei Teams gegeneinander berücksichtigt werden. Die entsprechende Tabelle sieht so aus:

					SSC Bos	SV Ober	1. FC Rie							
	Teilnehmer bzw. Mannschaft	Pkt	Diff.	erz. Tore										
3.	SSC Bosserode	6	5	5		2 : 0	3 : 0							
4.	SV Obersuhl	3	-1	2	0 : 2		2 : 1							
5.	1. FC Riedwald	0	-4	1	0 : 3	1 : 2								

Abb. 2.3

Die Betrachtung der in diesem zweiten Direktvergleich erreichten Punktzahlen ergibt bereits eine eindeutige Reihenfolge der drei Teams, so dass die Tordifferenzen und die erzielten Tore nicht mehr zur Unterscheidung herangezogen werden müssen.

2.1 Umsetzung des direkten Vergleichs in Formeln

Zur Erinnerung:

In Teil 1 hatten wir im Bereich T4:T12 Zahlen errechnet, die einfach zu den vorigen Rangzahlen im Bereich R4:R12 addiert werden und somit die neuen Rangzahlen im Bereich U4:U12 ergeben. Die ZÄHLENWENNS-Funktion in den Zellen T4:T12 führte eine Zählung in der Liste der Tordifferenzen aus dem Bereich J4:J12 durch.

Genau so wird es bei den direkten Vergleichen auch sein. Im Bereich L18:L26 steht wieder diese ZÄHLENWENNS-Funktion, deren Ergebnisse zu den vorigen Rangzahlen im Bereich I18:I26 addiert werden und somit die neuen Rangzahlen im Bereich M18:M26 ergeben.

Nur zählt die ZÄHLENWENNS-Funktion diesmal nicht in einer Liste der Tordifferenzen, sondern in einer Liste, die es noch gar nicht gibt, die also erst erschaffen werden muss. Und damit sind wir beim Hauptproblem und dem schwierigsten Schritt dieses Tutorials. Es geht um die angekündigte Matrixformel.

Wir brauchen eine Liste von erreichten Punkten. Für jedes Team muss geschaut werden, wie viele Punkte dieses Team in Begegnungen mit den punktgleichen Teams erreicht hat. Für die sechs punktgleichen Teams müssen also genau die Zahlen berechnet werden, die in Abb. 2.2 in der Spalte „Pkt“ zu sehen sind.

Diese Aufgabe übernehmen die Matrixformeln im Bereich K18:K26.

In der Zelle **K18** steht folgende Matrixformel:

```
{=SUMMENPRODUKT($O$31:$O$102*($J$31:$J$102=$D18)*ISTZAHL(VERGLEICH($K$31:$K$102;WENN($I$18:$I$26=$I18;$D$18:$D$26;""));0))+SUMMENPRODUKT($P$31:$P$102*($K$31:$K$102=$D18)*ISTZAHL(VERGLEICH($J$31:$J$102;WENN($I$18:$I$26=$I18;$D$18:$D$26;""));0))}
```

Sie setzt sich aus zwei gleichartigen Summanden zusammen:

```
{=SUMMENPRODUKT(...) + SUMMENPRODUKT(...)}
```

Wir betrachten zunächst den ersten Summanden:

```
SUMMENPRODUKT($O$31:$O$102*($J$31:$J$102=$D18)*ISTZAHL(VERGLEICH($K$31:$K$102;WENN($I$18:$I$26=$I18;$D$18:$D$26;""));0))
```

Der erste Teil ohne die ISTZAHL-Funktion

```
=SUMMENPRODUKT($O$31:$O$102*($J$31:$J$102=$D18))
```

addiert im Bereich O31:O102 alle Punktzahlen, bei denen links in der Spalte J der Name „1. FC Riedwald“ steht - also alle in Heimspielen erreichten Punkte des 1. FC Riedwald.

Damit diese Addition aller Punkte eingeschränkt wird auf diejenigen Punkte, die gegen die fünf punktgleichen Gegner erzielt wurden, muss innerhalb der Funktion SUMMENPRODUKT mit einem weiteren Faktor multipliziert werden.

Dieser dritte Faktor muss eine Liste von WAHR/FALSCH-Werten sein: WAHR dann, wenn es sich beim Gegner um ein punktgleiches Team handelt, FALSCH sonst. Diese Liste von WAHR/FALSCH-Werten wird von der Funktion ISTZAHL erzeugt.

2.2 Zerlegung der Funktion ISTZAHL in ihre Einzelteile

Betrachtet wird nun der Ausdruck

`ISTZAHL(VERGLEICH(K31:K102;WENN(I18:I26=$I18;$D$18:$D$26;""));0)`.

Wir gehen von innen nach außen vor. Zunächst die WENN-Funktion:

`WENN(I18:I26=$I18;$D$18:$D$26;"")`

Innerhalb der WENN-Funktion erzeugt die Bedingung `I18:I26=$I18` nicht nur *einen* Wahrheitswert (WAHR oder FALSCH), sondern eine ganze Liste (Matrix) von Wahrheitswerten. Diese Liste (Matrix) kann man sich anschauen, wenn man auf dem Tabellenblatt ‚Calc‘ den Bereich AR31:AR39 selektiert, in die Bearbeitungszeile oben die Formel `=I$18:I$26=I18` eingibt und mit STRG-UMSCHALT-EINGABE (Ctrl-Shift-Enter) abschließt.

Bei neueren Excel-Versionen ab Excel 2021 und Excel 365 kann man diese (Matrix-)Formel auch einfach in die Zelle V31 eingeben und mit der EINGABE-Taste abschließen. Es wird hier automatisch erkannt, dass es sich um eine Matrixformel handelt.

Die WENN-Funktion nutzt diese Liste von Wahrheitswerten, um aus der Liste aller Teams nur diejenigen herauszufiltern, bei denen die Bedingung erfüllt ist, also diejenigen, die in der Spalte I denselben Rang haben wie der 1. FC Riedwald. Um das Ergebnis der WENN-Funktion sichtbar zu machen, selektiert man den Bereich AS31:AS39, gibt die (Matrix-)Formel `=WENN(I18:I26=$I18;$D$18:$D$26;"")` ein und schließt mit STRG-UMSCHALT-EINGABE ab.

Im Bereich AR31:AS39 sieht man nun Folgendes:

WAHR	1. FC Riedwald
WAHR	FC Barcelona
WAHR	Eintracht Frankfurt
FALSCH	
FALSCH	
WAHR	Inter Mailand
FALSCH	
WAHR	SV Obersuhl
WAHR	SSC Bosserode

Abb. 2.4

Die Funktion SUMMENPRODUKT soll alle Punktzahlen im Bereich O31:O102 addieren, wenn in Spalte J das Team „1. FC Riedwald“ steht und in der Spalte K ein Team aus der Liste punktgleicher Teams (siehe Abb. 2.4). Wir brauchen deshalb eine weitere Liste von 72 Wahrheitswerten WAHR/FALSCH, die für jedes Team aus dem Bereich K31:K102 ein WAHR enthält, wenn es eines der sechs punktgleichen Teams ist, ansonsten ein FALSCH.

Dazu ermittelt die Funktion VERGLEICH für jedes Team aus dem Bereich K31:K102 den Index in der Liste der sechs punktgleichen Teams. Die VERGLEICH-Funktion erzeugt bei jedem Team, das nicht in der Liste der punktgleichen Teams enthalten ist, den Fehler #NV.

Die Liste von Indizes, die von der Funktion VERGLEICH generiert wird, können wir ebenfalls sichtbar machen: Wir selektieren den Bereich AT31:AT102, geben in die Eingabezeile oben die (Matrix-)Formel `=VERGLEICH(K31:K102;WENN(I18:I26=$I18;$D$18:$D$26;""));0` ein und schließen die Eingabe mit STRG-UMSCHALT-EINGABE ab (Ergebnisse siehe Abb. 2.5).

Die Funktion ISTZAHL wandelt jeden Fehlercode #NV in FALSCH und jeden Index in WAHR um.

Die Liste dieser 72 WAHR/FALSCH-Werte können wir sichtbar machen, indem wir den Bereich AU31:AU102 selektieren, in die Bearbeitungszeile oben die (Matrix-)Formel **=ISTZAHL(VERGLEICH(\$K\$31:\$K\$102;WENN(\$I\$18:\$I\$26=\$I18;\$D\$18:\$D\$26;"");0))** eingeben und die Eingabe mit STRG-UMSCHALT-EINGABE abschließen.

WAHR	1. FC Riedwald	2	WAHR
WAHR	FC Barcelona	8	WAHR
WAHR	Eintracht Frankfurt	#NV	FALSCH
FALSCH		6	WAHR
FALSCH		1	WAHR
WAHR	Inter Mailand	#NV	FALSCH
FALSCH		3	WAHR
WAHR	SV Obersuhl	#NV	FALSCH
WAHR	SSC Bosserode	8	WAHR
		9	WAHR
		2	WAHR
		#NV	FALSCH
		1	WAHR
		8	WAHR
		#NV	FALSCH
		3	WAHR
		6	WAHR
		#NV	FALSCH
		#NV	FALSCH
		9	WAHR
		1	WAHR

Abb. 2.5

Zwischenbemerkung:

Die vier Listen in Abb. 2.5 werden beim Auswerten der Matrix-Formeln von Excel intern (unsichtbar) erzeugt und gleich weiterverarbeitet, ohne dass sie als Zwischenergebnisse auf dem Blatt abgelegt werden müssen. Dadurch erhöht sich die Performance.

Aber beim Einsatz von Matrixformeln gibt es noch einen zweiten Aspekt:

Die hier dargestellten vier Listen gelten nur für die erste SUMMENPRODUKT-Funktion der Matrixformel in Zelle K18. Für die zweite SUMMENPRODUKT-Funktion und für die Matrixformeln in anderen Zellen können die vier Listen wieder ganz anders aussehen. Für die 81 Matrixformeln auf dem Tabellenblatt ‚Calc‘ werden 81 x 8 = 648 dieser Listen benötigt. Sie alle als Zwischenergebnisse anzuzeigen ergäbe einen großen und unübersichtlichen Formelsalat.

2.3 Abschluss des ersten Direktvergleichs

In der Formel **{=SUMMENPRODUKT(...)+SUMMENPRODUKT(...)}** gibt es zwei gleichartige Summanden. Der erste addiert alle Punktzahlen, die das betreffende Team als Heim-Team gegen punktgleiche Teams erzielt hat. Der zweite Summand addiert alle Punktzahlen, die das betreffende Team als Gast-Team gegen punktgleiche Teams erzielt hat.

Wir haben nun im Bereich K18:K26 die Liste der Punktzahlen, die jedes der sechs punktgleichen Teams gegen die anderen erreicht hat (vergleiche Abb. 2.2, Spalte „Pkt“).

Die ZÄHLENWENNS-Funktion im Bereich L18:L26 zählt nun für jedes Team, wie viele punktgleiche Teams eine höhere Punktzahl aus dem direkten Vergleich erzielt haben. Beim Team „Inter Mailand“ steht eine 5, weil Inter Mailand nur 3 Punkte im direkten Vergleich erzielt hat, die anderen 5 Teams jedoch jeweils 6 Punkte. Somit fällt Inter Mailand auf den Rang 7 zurück.

Für die anderen 5 Teams wird im nächsten Schritt geprüft, wie die Tordifferenzen aus dem direkten Vergleich aussehen.

Im Bereich O18:O26 arbeitet die gleiche Matrixformel wie im Bereich K18:K26, nur ermittelt sie jetzt die Tordifferenzen der sechs punktgleichen Teams. Es werden also genau die Zahlen berechnet, die oben in der Abb. 2.2 in Spalte „Diff.“ stehen.

Die Formeln im Bereich P18:P26 zählen nun für jedes Team, wie viele der immer noch auf demselben Rang stehenden Teams eine bessere Tordifferenz haben. Bei den fünf Teams, die noch auf Rang 2 stehen, fallen drei auf Rang 3 zurück und ein Team auf Rang 6. Der FC Barcelona mit der besten Tordifferenz von +3 verbleibt als einziges Team auf Rang 2.

Im nächsten Schritt (Formeln im Bereich S18:U26) werden nach dem gleichen Mechanismus die erzielten Tore derjenigen Teams geprüft, die immer noch auf demselben Rang stehen (Rangliste im Bereich Q18:Q26).

Die drei Teams auf Rang 3 unterscheiden sich allerdings auch bei den erzielten Toren nicht. Die Rangzahlen ändern sich bei diesem Schritt somit nicht.

2.4 Zweiter Direktvergleich

Beim zweiten Direktvergleich geht es nun nach dem bekannten Muster weiter. Der entscheidende Unterschied ist, dass sich die Matrixformeln des ersten Direktvergleichs alle auf die Rangzahlen im Bereich I18:I26 beziehen, die Matrixformeln des zweiten Direktvergleichs dagegen auf die Rangzahlen im Bereich U18:U26.

Die Matrixformeln des zweiten Direktvergleichs berechnen die Zahlen in Abb. 2.3, Spalten „Pkt“, „Diff.“ und „erz. Tore“.

Bereits im ersten Schritt dieses zweiten Direktvergleichs, im Vergleich der erreichten Punktzahlen, werden die endgültigen Rangzahlen der verbliebenen drei Teams entschieden. Der Vergleich der Tordifferenz und der erzielten Tore hat dann keine weitere Auswirkung mehr.

Teil 3

Weitere Bemerkungen zur Excel-Datei

Nach dem zweiten Direktvergleich folgen dann die Kriterien

- Tordifferenz aus allen Gruppenspielen
- erzielte Tore aus allen Gruppenspielen
- Tie-Break (Fair-Play, Weltrangliste oder Los).

Die finalen Rangzahlen stehen für den Modus 1 im Bereich E4:E12 und für den Modus 2 im Bereich E18:E26. Die Auswahl auf dem Tabellenblatt ‚Einstellungen‘ entscheidet dann, welche diese beiden Listen im Bereich A4:A12 erscheint.

Aus diesem Bereich A4:A12 holen sich die Formeln der Tabelle auf dem Tabellenblatt ‚Ergebnisse‘ die aktuellen Rangzahlen, so dass diese Tabelle die Rangreihenfolge entweder nach Modus 1 oder nach Modus 2 anzeigt, je nachdem, was in den Einstellungen gewählt wurde.

